# Maintainer: Sergej Pupykin <pupykin.s+arch@gmail.com>
# Maintainer: Patryk Kowalczyk < patryk at kowalczyk dot ws>

pkgname=spice-protocol
pkgver=0.14.3
pkgrel=1
pkgdesc="Headers for SPICE protocol"
arch=(any)
url="https://spice-space.org"
license=('BSD' 'LGPL2.1')
makedepends=('meson')
source=(https://gitlab.freedesktop.org/spice/spice-protocol/-/archive/v$pkgver/spice-protocol-v$pkgver.tar.gz)
md5sums=('d11f4f037f667226c9750c62976bdb32')

build() {
  mkdir build
  cd "$srcdir/$pkgname-v$pkgver"
  arch-meson ../build
  ninja -v -C ../build
}

package() {
  cd "$srcdir/$pkgname-v$pkgver"
  DESTDIR="$pkgdir/" ninja -C ../build install
  install -Dm644 COPYING "$pkgdir/usr/share/licenses/$pkgname/COPYING"
}
